

Методические указания для практикума по обработке эшелле спектров в пакете MIDAS

Все действия проводятся в терминале в директории `work/Echelle`
если вы в другой, то выполните команду `cd work/Echelle` или `cd ~/work/Echelle`

Для начала работы в пакете MIDAS служит команда:

```
inmidas
```

либо, если требуется продолжение предыдущей сессии MIDAS

```
gomidas
```

Внутри MIDAS можно получить информацию по командам и их аргументам с помощью команды `help`, например:

```
help indisk/fits
```

Создаем каталог спектров объекта. Усредняем все кадры с помощью медианной фильтрации.

! Внимание медианную фильтрацию можно применять только для кадров, снятых с одной экспозицией. Иначе нужно обрабатывать каждый кадр отдельно и складывать только после удаления космиков. Если космики не влияют, то можно вместо `median` написать `average` для усреднения кадров

! Знак `&` говорит о том, что файл будет временным (в файловой системе он начинается с `middummy`), и по завершении сессии может быть удален.

```
create/icat obj      obj*.fit
average/ima &obj    = obj.cat  ?? median
loa/ima &obj sc=-3
```



Создаем каталог спектров плоского поля. Также усредняем все кадры с помощью медианной фильтрации.

```
create/icat ff      flat*.fit
average/ima &ff    = ff.cat  ?? median
```

```
loa/ima &ff sc=-3
```




То же самое для кадров BIAS:

```
create/icat bias bias*.fit
average/ima bias = bias.cat ? ? median
```

То же самое для кадров калибровочной лампы, предварительно преобразовав интенсивности из диапазона -32768 ... 32768 в диапазон 0 ... 65536 (это особенности конкретной электроники, значения от 0 до 32768 соответствуют реальности, при превышении используется диапазон от -32768 до 0; Эту процедуру мы не делали для других кадров поскольку там все значения не превышали 32768):

```
REPLACE/IMAGE thar01.fit thar01 <,0.0=thar01.fit+65536
REPLACE/IMAGE thar02.fit thar02 <,0.0=thar02.fit+65536
```

```
create/icat thar thar0[1-2].bdf
average/ima &thar = thar.cat ? ? average
loa/ima &thar sc=-3 cu=f,i
```



Вычитаем BIAS из всех кадров:


```
comp/ima ff      = &ff      - bias
comp/ima thar    = &thar    - bias
comp/ima obj     = &obj     - bias
```

Переходим к обработке эшелле спектров, подключив пакет echelle:

```
set/cont echelle
```

Первым делом определяем положения отдельных эшелле порядков (будет найден 41 порядок):

```
DEFINE/HOUG ff 0 0 CENTER
```




Проверить качество положений порядков на любом эшелле кадре при любом изменении масштаба можно командой:

```
loa/ech
```

Посмотрим вертикальный разрез по кадру:

```
cre/gra
plo/col ff 500
```




Увеличив масштаб по оси X:

```
plo/col ff 500 400,500
```

командой `get/gcur` определим длину щели (это соответствует ширине порядка на выводимом графике). Получается около 10 пикселей. Выделяем эшелле порядки спектра калибровочной лампы в отдельные спектры:

```
extr/ech thar thar_ext 10  
loa/ima thar_ext sc=-3,10 cu=f,i
```



Из исходного кадра размером 1024x1024 получается кадр размером 1024x41, то есть каждая строка соответствует одному порядку.

Далее выполним калибровку по длинам волн. Для этого нужно найти эмиссионные линии в спектре лампы, что в свою очередь требует знания типичной толщины линии и минимальной интенсивности, поэтому строим график части одного порядка, например 10-ого (можно погулять по порядкам и посмотреть):

```
plo/row thar_ext 10 200,300
```

Получаем толщину около 4 пикселей, минимальная интенсивность около 100. Запускаем команду поиска линий:

```
sear/ech thar_ext 4,100
```

Найдено 864 линии. В среднем по 20 на порядок. Создается таблица линий `line.tbl`.

Начинаем идентификацию линий и построение дисперсионного решения по эшелле кадру. Для идентификации пользуемся списком линий из таблицы `thar100.tbl`, в каждом порядке аппроксимируем положения линий полиномом 3-ей степени, отбрасываем все линии, которые отскакивают больше, чем на 0.01 Å, поиск соответствия линии ведем в интервале 0.1 Å, используем метод PAIR (необходимо правильно отождествить на кадре две пары линий, в каждая пара — это одна и та же линия, располагающаяся в двух соседних порядках — смотрите рисунок в конце документа). Отмечаем сразу все линии с помощью квадратного курсора. Сначала первую пару, одну линию с левой стороны кадра и ее же в соседнем порядке с правой стороны кадра. Также и вторую пару. Изменение размеров курсора с помощью клавиш стрелок. Рекомендуется выбрать линии 6049.0517 (94 порядок справа и 93 - слева) и

6114.9234 (93 порядок справа и 92 — слева). После окончания выбора выход правой клавише мышки. Команда запросит номер абсолютного порядка первой линии и ее длину волны, потом только длину волны второй пары линии:


```
loa/ima thar cu=f,i  
IDENTI/ECHE thar MID_ARC:thar100 3 -0.01 -0.1 pair
```

Вывод команды покажет диапазон длин волн каждого порядка и среднюю ошибку длин волн. Желательно, чтобы не менее 50% (это примерно) линий были использованы в идентификации. Сохраняем все нужные параметры в сессию с именем «ту»

```
save/ech tu
```


При необходимости улучшить калибровку можно повторить команду `sear/ech` с другими параметрами, чтобы увеличить число возможных линий или уменьшить, так как могут попадаться «ложные» отождествления из-за шумов. После этого команда идентификации не требует ввода длин волн и будет запускаться с другим последним параметром (`guess tu`), беря информацию из сохраненной сессии `tu`:

```
IDENTI/ECHE thar MID_ARC:thar100 3 -0.01 -0.1 guess tu
```



Удаляем рассеянный свет из спектра звезды с помощью полинома и выводим на графическое окно:

```
SUBTRA/BACK obj bg objb POLY yes
```




Изменить степень полинома (по осям X и Y) можно командой

```
set/ech BKGPOL=4,5
```

Определяем положение участков со спектром неба. Для этого появившимся курсором ткните в центр кадра на один из порядков, а затем сразу в графическом окне отметьте точки, в которых, как вы полагаете, присутствует свет неба.

```
cle/dis  
define/sky objb 2 cursor
```




и вычитаем его из спектра звезды:

```
extract/sky objb sky filter  
set/ech EXTMTD=average  
extract/ech objb obj_ext  
comp/ima obj_sky = obj_ext - sky
```


Смотрим на 10 эшелле порядок в спектре звезды:

```
plo/row obj_sky 10
```



Для учета функции блеска эшелле («выпрямление» спектра) используем спектр плоского поля, который позволит нам учесть и мелкую неоднородность. Учитываем рассеянный свет, выделяем порядки из спектра плоского поля.

```
SUBTRA/BACK ff bg ffb POLY
EXTRAC/ECHE ffb ff_ext
plo/row ff_ext 10
```



Находим среднее значение интенсивности спектра плоского поля:

```
STAT/IMA ff_ext
```


```
frame: ff_ext (data = R4)
complete area of frame
minimum, maximum: 0.000000e+00 1.625212e+04
at pixel (1,1),(456,34)
mean, standard_deviation: 9.160066e+03 3.912847e+03
3rd + 4th moment: -0.372653 2.13769
total intensity: 3.84576e+08
exact median, 1. mode, mode: 9.382761e+03 3.186691e+01 8.699666e+03
total no. of bins, binsize: 256 6.373382e+01
# of pixels used = 41984 from 1,1 to 1024,41 (in pixels)
```

и нормируем его

```
COMPUT/IMAG ffn = ff_ext/9160
```

Выполняем учет неоднородностей плоского поля для спектра звезды:

```
comp/ima obj_norm = obj_sky/ffn
plo/row obj_norm 10
```




Видим, что спектр почти выровнен, но не полностью, так как распределения энергии в спектрах звезды и лампы плоского поля различны. Но этого достаточно, чтобы в дальнейшем провести уровень непрерывного спектра. Если требуется точность, то наблюдают звезду-стандарт с известным распределением энергии в спектре.

Применяем калибровку по длинам волн к спектру звезды:

```
rebin/echе obj_norm obj_reb
```

Соединяем все эшелле порядки в единый спектр:

```
merge/ech obj_reb obj_1D  
plo/row obj_1D
```




Волны отражают неравномерность спектральной чувствительности ПЗС матрицы. Сохраняем в FITS-формат для дальнейшего анализа в других программах:

```
outdis/fits obj_1D obj_1D.fits
```

Вырежем небольшой участок спектра около линий натрия:

```
EXTRAC/IMAG na = obj_1D [5860,1:5930,1]  
plo/row na
```




Вручную отметим точки непрерывного спектра, где отсутствуют сильные линии, и следующая команда построит нам континуум cont:

```
NORMALIZE/SPECTRUM na cont
```

И отнормируем спектр на уровень континуума:

```
comp/ima norm = na/cont  
plo/row norm
```



Измерим положение любых линий, выделяя левый и правый края курсором:

```
CENTER/GAUS GCURSOR ? ABS
```

